
Conversion Factors
Inch/Pound to SI

Multiply By To obtain
Length

inch (in.)  25.4 millimeter (mm)
mile (mi)  1.609 kilometer (km)

Energy
Langleys per day (Lg/d)  279.12 Watts per square meter (W/m2)

Temperature in degrees Fahrenheit (°F) may be converted to degrees  Celsius (°C) as follows:

      °C=(°F-32)/1.8



Comparison of Potential Evapotranspiration Calculated 
by the LXPET (Lamoreux Potential Evapotranspiration) 
Program and by the WDMUtil (Watershed Data 
Management Utility) Program

By Elizabeth A. Murphy

Abstract
Potential evapotranspiration can be estimated using observed meteorological data, and many methods have been 

developed to accomplish this estimation.  This report, prepared by the U.S. Geological Survey in cooperation with 
DuPage County Department of Engineering, Stormwater Management Division, describes the LXPET (Lamoreux 
Potential Evapotranspiration) program and compares the daily and hourly values calculated by LXPET with the 
values calculated by WDMUtil (Watershed Data Management Utility).  The inputs to both LXPET and WDMUtil 
are total daily solar radiation in Langleys per day, total daily wind movement in miles, and average daily dewpoint 
temperature in degrees Fahrenheit.  Average air temperature in degrees Fahrenheit is used in LXPET, whereas either 
maximum and minimum daily air temperature or average daily air temperature in degrees Fahrenheit is used in 
WDMUtil.  The latitude (degree, minute, second) where evaporation is calculated is a required WDMUtil input.

The daily evaporation time-series data generated by LXPET are smaller by a factor of 0.7 than the daily time-
series data generated by WDMUtil.  The 0.7 factor is the same as the 0.7 correction factor commonly applied to 
pan evaporation data to approximate lake evaporation.  The time series from either program can be used in hydro-
logic modeling depending on the expected model input. The Hydrologic Simulation Program-Fortran (HSPF), the 
hydrologic model used in this study, uses potential evapotranspiration as input.  Because lake evaporation is used 
as potential evapotranspiration in hydrologic modeling, LXPET time-series values should be used as HSPF input or 
the WDMUtil time-series values should be multiplied by 0.7 before being input into HSPF.  However, because the 
WDMUtil time series is adjusted for season of the year, this time series may give more realistic evaporation values 
when HSPF is run over multiple seasons.

Introduction
The U.S. Geological Survey (USGS), in cooperation with DuPage County Department of Engineering, Storm-

water Management Division, has been conducting hydrologic and hydraulic modeling studies in DuPage County, Ill. 
since 1997 using Hydrologic Simulation Program-Fortran (HSPF) as the hydrologic model.  Potential evapotranspi-
ration is an important parameter in hydrologic modeling, and it can be estimated using observed meteorological data.  
A study was begun in 2003 to compare the potential evapotranspiration time series calculated by LXPET (Lamoreux 
Potential Evapotranspiration) and WDMUtil (Watershed Data Management Utility), a program packaged with HSPF 
as part of the BASINS (Better Assessment Science Integrating point and Nonpoint Sources) software system (U.S. 
Environmental Protection Agency, 2001).  The purpose of this report is to compare the evapotranspiration time series 
generated by LXPET with that generated by WDMUtil.  The differences between the two time series are discussed, 
as well as the time series used for HSPF model simulation.  The differences between the two time series are also 
shown graphically using meteorological inputs from the Argonne National Laboratory in Illinois.  This meteorology 
station has been operated since 1948 and data from January 1, 1990, to September 30, 2003, are used in this study 
(Argonne National Laboratory, 2005).



2  Comparison of Potential Evapotranspiration from the LXPET Program and WDMUtil 

The USGS was given a program for calculating potential evapotranspiration based on the method of Kohler and 
others (1955) as adapted to digital computers by Lamoreux (1962) by the Northeastern Illinois Planning Commission 
(NIPC).  The USGS also was given a program for distributing the calculated daily time series to hourly values (Tom 
Price, Northeastern Illinois Planning Commission, presently with Conservation Design Forum, written commun., 
1997).  The program code probably originated as part of the LANDS module of the HSP (Hydrocomp Simulation 
Program) hydrologic model (Hydrocomp International, Inc., 1970) and was coded in PL/1 before it was ported to 
FORTRAN (FORmula TRANslation) in the early 1980s.  The program was coded in FORTRAN by Hey and Asso-
ciates under contract to NIPC (Tom Price, Northeastern Illinois Planning Commission, presently with Conservation 
Design Forum, written commun., 2005). Kohler and others (1955) developed graphs to determine lake evaporation 
from air temperature, dewpoint temperature, solar radiation, and wind movement.  Lamoreux (1962) developed 
equations from the graphs, so a computer can be used to calculate the evaporation.  The USGS modified and com-
bined the two programs to automatically distribute the daily time series to an hourly time series and to accept aver-
age daily air temperature as an input time series rather than a maximum daily air temperature and a minimum daily 
air temperature.  The USGS named the program LXPET to reflect the use of the Lamoreux methods and calcula-
tion of potential evapotranspiration.  No complete documentation is available for LXPET; however, this report will 
describe the equations and methods used by LXPET to calculate potential evapotranspiration and provide instruc-
tions for using LXPET in appendix A.  The FORTRAN source code for the LXPET program has been included in 
appendix B.  A digital version of the FORTRAN source code and the compiled LXPET program can be requested 
from the USGS Illinois Water Science Center.

Comparison of Potential Evapotranspiration Equations
LXPET and WDMUtil can both be used to calculate evaporation. LXPET is based on the method of Kohler and 

others (1955) as adapted to digital computers by Lamoreux (1962). LXPET calculates lake evaporation.  The Pen-
man pan evaporation calculations used in WDMUtil are based on the Penman (1948) formula and the method of 
Kohler and others (1955).

The inputs to LXPET are total daily solar radiation in Langleys per day (Lg/d), total daily wind movement in 
miles (mi), average daily dewpoint temperature in degrees Fahrenheit (°F), and average daily air temperature (°F).  
Average air temperature (°F) is required in LXPET, whereas WDMUtil requires either maximum and minimum daily 
air temperatures, or average daily air temperature (°F).  WDMUtil also requires the latitude (degree, minute, second) 
where evaporation is calculated.  The output of LXPET is an hourly lake evaporation time series (thousandths of an 
inch).  The output of WDMUtil is an hourly pan evaporation time series (inches).

The LXPET program contains the following equations for the calculation of lake evaporation (EVAP). 

EVAP = B * A * 1,000 ,           (1)

where

B = [ 0.015 + ((Ta + 398.36)-2) * 68,554,000,000 * exp (    -7,482.6     )]-1 

,
                                                                                   (Ta + 398.36)

A = exp ((Ta – 212) * (0.1024 – 0.01066 * ln(R))) – 0.0001 + 0.0105 * C 0.88 * (0.37 + 0.0041 * Up) ,

where

C = 6,413,300 * [exp (    -7,482.6    ) – exp (        -7,482.6         )] ,
                                     (Ta + 398.36)               (DEWP + 398.36)



where Ta is the air temperature in °F, R is the total daily solar radiation in Langleys, Up is total daily wind movement 
in miles, and DEWP is the dewpoint temperature in °F.

The WDMUtil Version 2.0 Users’ Manual (2001) states the Penman daily pan evaporation formula as

E =  
(Q * DEL + Ea * GAM)          

(2)––––––––––––—–––––  ,
        (DEL + GAM)

where E is pan evaporation in inches, Q is net radiation exchange, DEL is the slope of the saturation vapor-pressure 
curve at the given air temperature, GAM is 0.0105 inch Hg/°F (defined by Bowen’s ratio), and E

a
 is pan evaporation 

when air temperature equals water temperature.

The empirical equation for Q * DEL is

Q * DEL = exp [(Ta – 212)(0.1024 – 0.01066 * ln(R))] – 0.0001.

(The last term in the equation above is actually listed as 0.000 in the WDMUtil documentation (WDMUtil Version 
2.0 Users’ Manual, 2001), but that was confirmed to be an error (Paul Cocca, U.S. Environmental Protection Agency, 
written commun., 2004.))

The equation for Ea is

Ea = (0.37 + 0.0041 * Up) * (es – ea)0.88,

where (es – ea) is the vapor-pressure deficit between the surface and dewpoint temperatures.

As is shown in table 1 and discussed below, the key difference between the LXPET and WDMUtil 
programs is the slope of the saturation vapor-pressure curve, DEL.  

DEL is calculated in LXPET with the equation

DEL = (Ta + 398.36)-2 * 68,554,000,000 * exp (    -7,482.6     ) .    (3)
                                                                     (Ta + 398.36)

DEL is calculated in WDMUtil with the equation

DEL = (      7,482.6      ) exp (15.674 –      7,482.6    ) .      (4)
       (Ta + 398.36)2

                           (Ta + 398.36)

A side-by-side comparison of the LXPET and WDMUtil evaporation equations is provided in table 1.  This 
table illustrates that the programs are based on the same set of equations. The only difference between pan evapora-
tion and lake evaporation is in the equation for DEL. 

Comparison of Potential Evapotranspiration Equations  3

The Lake-Correction Factor

When the daily-evaporation time series from these two programs were compared, the LXPET values were 
smaller than the values calculated with the WDMUtil program.  When the LXPET values were divided by the 
WDMUtil values, the result always was approximately 0.7.  A double-mass curve with a constant line of 0.7 is 
depicted in figure 1 to show the consistency of the 0.7 factor over the period January 1990-September 2003.  

The DEL equation in the LXPET calculations can be rewritten to reveal a 0.7 term not present in the WDMU-
til DEL equation (table 1). This 0.7 factor is described as a lake correction by Kohler and others (1955) and is used 
to convert from Class A pan evaporation to lake evaporation.  According to Viessman and others (1989, p. 102), 
“most investigators have assumed that potential evapotranspiration is equal to lake evaporation as determined from 
National Weather Service Class A pan records.”  However, Viessman and others (1989) also explain that the potential 
evapotranspiration is a function of soil type and vegetation, so the simple use of lake evaporation is not theoretically 
correct.

Hourly Distribution 

The LXPET program calculates daily lake evaporation values and distributes the daily values to hourly values.  
The distribution process is incorporated into the program and is not seen by the user as a separate step.  In contrast 
to the WDMUtil method described below, the LXPET distribution process does not change seasonally.  The LXPET 
program always distributes daily evaporation over the hours from 6:00 to 18:00.

The WDMUtil hourly pan evaporation time-series calculation takes two steps.  The daily evaporation time series 
is calculated, then must be disaggregated to an hourly time series as a second step.  WDMUtil requires the latitude 
of the site as an input for the disaggregation process.  The seasons at the site are determined based on the latitude.  
WDMUtil disaggregates the evaporation differently depending on season.  The winter evaporation is spread over 
fewer hours (for example, 7:00-15:00) than the spring or fall (for example, 5:00-17:00) with evaporation occurring 
over the most hours in the summer (for example, 4:00-18:00).  This disaggregation method more realistically divides 
the daily evaporation over the hours of the day than the LXPET method, which does not change seasonally. 

The difference in the disaggregation on a winter day and a summer day at the meteorological observation station 
at Argonne National Laboratory in northeastern Illinois is illustrated in figure 2.  In winter, LXPET peak evaporation 
is lower than WDMUtil, whereas in summer, LXPET peak evaporation is higher than WDMUtil.  LXPET distributes 

Figure 1.  LXPET (Lamoreux Potential Evapotranspiration) and WDMUtil (Watershed Data Management Utility) double-mass curve 
and the 0.7 factor for daily evaporation calculated at Argonne National Laboratory in Illinois for the period January 1, 1990,–Septem-
ber 30, 2003.

Comparison of Potential Evapotranspiration Equations  5

the evaporation over more hours in winter than WDMUtil.  In summer, LXPET distributes the evaporation over 1 
hour less than WDMUtil. 

Hydrological Simulation Program-Fortran (HSPF)
The HSPF program uses potential evapotranspiration as a model input.  The following discussion of the appro-

priate HSPF input time series is taken from the HSPF manual (Bicknell and others, 1997) regarding the input of 
potential evapotranspiration to the model.

“ET potential or demand is supplied as an input times series, typically using U.S. Weather Bureau Class A pan 
records plus an adjustment factor. The data are further adjusted for cover in the parent subroutine PWATER.”

This data description indicates that pan evaporation time series should be adjusted before it is input to HSPF.  
A conclusion of the present study is that either the LXPET time series should be used or the WDMUtil time series 
should be multiplied by 0.7 before it is input into HSPF.

Summary
The U.S. Geological Survey, in cooperation with DuPage Department of Engineering, Stormwater Manage-

ment Division, has compared the evaporation time series calculated by LXPET and WDMUtil.  Either program can 
be used to calculate an evaporation time series for hydrologic modeling.  The programs use similar equations and 
require similar input data.  The LXPET program generates a lake evaporation time series, whereas WDMUtil gener-
ates a pan evaporation time series.  The difference between the programs is the equation used for DEL, the equation 
of the saturation vapor-pressure curve. The LXPET daily values are adjusted by a lake correction factor of approxi-
mately 0.7 compared to the WDMUtil daily values of pan evaporation.  

The LXPET daily evaporation values are disaggregated to hourly values in a uniform manner that does not 
change based on season of the year or latitude.  WDMUtil uses latitude to alter the disaggregation from daily to 
hourly values based on seasonal and regional differences. 

HSPF is a hydrologic model used to simulate the amount of runoff during storms in watersheds.  The HSPF 
program requires potential evapotranspiration as an input time series.  Because lake evaporation is used by HSPF 
modelers as potential evapotranspiration, either LXPET time-series values should be used as HSPF input or the 
WDMUtil pan evaporation time-series values should be multiplied by 0.7 before being input into HSPF.  However, 
because the WDMUtil time series is adjusted for season of the year and latitude, multiplying it by 0.7 may give more 
realistic potential evapotranspiration values than the LXPET output time series when the HSPF model is run over 
multiple seasons.

Figure 2.  Evaporation distribution by LXPET (Lamoreux Potential Evapotranspiration) and WDMUtil (Watershed Data Management 
Utility) at Argonne National Laboratory in Illinois on A) January 15, 1995 (winter), and B) July 15, 1995 (summer).



The FORTRAN source code for the LXPET program has been included in this report.  A digital version of the 
FORTRAN source code and the compiled LXPET program can be requested from the USGS Illinois Water Science 
Center.

Appendix A:  Computing evapotranspiration time series using LXPET (Lamoreux 
Potential Evaporation)

LXPET is a program that computes an hourly potential evapotranspiration (lake evaporation) time series (EVAP) 
using daily average dewpoint temperature (DEWP), daily average temperature (TEMP), daily total wind movement 
(WIND), and daily total solar radiation (SRAD) data files as input.  The use of LXPET with input data files cre-
ated in the GenScn (Generation and Analysis of Model Simulation Scenarios) program (Kittle and others, 1998) is 
described below, but the input data files can be from any source as long as they are properly formatted.  Instructions 
for generating input data files using GenScn, correctly formatting these data files, and using LXPET as a DOS appli-
cation are given in the steps below.  Tips on troubleshooting common problems also are included.

Obtaining the data from the WDM (Watershed Data Management) with the Generation and Analysis 
of Model Simulation Scenarios (GenScn) Program 

Note: These instructions are based on GenScn version 2.3 (Kittle and others, 1998).  Certain details may 
vary in other versions.

(1) Open GenScn and the WDM file with the DEWP, SRAD, TEMP, and WIND data. 

(2) Select OBSERVED, the constituent (for example, DEWP), and the location (for example, Argonne).  Add this 
 combination to your Time Series frame (see figure A1 below).

(3) Enter the beginning and ending dates needed for the computed EVAP.

(4) From the T step, Units menu, select the appropriate units for the data.  “Sum” should be selected for SRAD and  
 WIND, “Average” for DEWP and TEMP.

(5) Once the T step unit is chosen, another drop-down box will appear.  Make sure “1 Day” appears.  If not, select  
 “Day” from the drop-down box.

Figure A1.  The Generation and Analysis of Model Simulation Scenarios (GenScn) program with the Argonne National Laboratory in 
Illinois dewpoint time series selected.

(6) Select the List option (second from the left in the Analysis menu at the bottom of the screen). 

(7) The date format should be 4-digit year, space, 2-digit month, space, 2-digit day.  If the data are not in this for  
 mat, select edit and date format to adjust the format of the date.

(8) Select File and Save to Text File.  Save the text file as a space delimited file.  Make sure that the data files are  
 saved in the same directory as the LXPET program.  Choose a file name with no more than eight characters and  
 an extension with no more than three characters.  This file name format is necessary because the data file will be  
 used on a DOS platform.

Repeat steps 4-7 for the other constituents.

Formatting the data for LXPET

(1) Open the data file in a text editor.

(2) Delete the header lines of the file (lines at the top that contain information about the data set).  These header   
 lines will cause an error message from LXPET. 

(3) Make sure there are 21 spaces between the last character of the date and the first character of the value for   
 that date. The first character of the value may be a minus sign in the case of temperature and dewpoint files.    
 Make sure there are no tabs, only spaces.  The LXPET program does not accept tab characters.  An example   
 of a properly formatted file (with the spaces visible) is included as figure A2 below.

(4) Save the changes.

Repeat these steps for each of the other data files.

Running LXPET

(1) Make sure that LXPET and the four input data files are in the same directory.

(2) Run LXPET by either double-clicking on LXPET.exe from the Windows Explorer or typing LXPET at the   
 prompt from a DOS screen or window from its directory.  In the DOS window, change to the directory where  
 LXPET and the data files are located (see figure A3 below).  No parameters need to be added to LXPET at the  
 command line.  It is preferable to start from a DOS window, so the error messages are shown in the DOS win 
 dow if LXPET does not work correctly.  If the LXPET.exe is opened by double-clicking from Windows, the   
 DOS window will close automatically if an error occurs.

(3) When prompted, enter the name of the solar radiation data file including the extension.  Hit enter when fin  
 ished.

(4) Next, the multiplication factor for solar radiation is requested.  Type in 1 for this multiplication factor.  Use 1  
 as the multiplication factor for the other data types as well.  

(5) Enter the names of the other data files as prompted by the program.

(6) The last step prompts for the name of the evaporation file.  This file is where LXPET saves the computed   
 EVAP time series.  Choose a name no longer than eight characters and an extension of no more than three char 
 acters.  It is recommended that .hsp is used as the extension. The .hsp format is the default output format and  
 will make it easier to input the results to a WDM file using HSPF.

Open the EVAP file in a text editor.  If a blank file, all zeros, or some other unexpected output is listed, then 
refer to the troubleshooting tips below.

Figure A3.  The LXPET (Lamoreux Potential Evapotranspiration) program, in a DOS window, with all the prompts displayed.

Figure A2.  An example of a properly formatted input file of the dewpoint data for LXPET (Lamoreux Potential Evapotranspiration).

Troubleshooting Tips

Problem: The output file shows 24 sets of 24 values for each day.

Cause: This problem is caused by using hourly data instead of daily data.  LXPET treats each hour as a day.  
To fix this problem, go back into GenScn and retrieve daily data.  Rerun LXPET with daily data.

Problem: The output file shows 0.00 for all the values.

Cause: This type of error could be caused by entering the wrong file name for each data type (for example, 
the file name of the wind data was entered when the program prompted for solar radiation data).  Rerun LXPET and 
make sure the right filename is entered for each data type.  If this solution doesn’t work, check that the data files are 
named correctly.  Another potential cause of this problem is that one or more of the data files were not formatted cor-
rectly.  As mentioned earlier, there should be 21 spaces between the last character of the date and the first character 
of the value.  

Problem: Evaporation values that look appreciably low.

Cause: One of the data files was not formatted correctly.  The LXPET program does not use improperly 
formatted data files in its calculations; therefore, the evaporation would be lower than the correct value.

 
Problem: Output for evaporation has 0.00 for the beginning and end of the day and ****** characters for the 

rest of the day.

Cause: This problem is caused by the evaporation value being calculated as too high; for example, if 11 
were typed instead of 1 for the multiplication factor for DEWP, TEMP or SRAD.  Be aware that if this problem 
occurs from an improperly formatted wind data file, the evaporation values would be larger than the correct values 
but not high enough to make it obvious that a problem is present.

Problem: After the program is run, an error message results as follows:

Invalid decimal character * was detected (unit=**)  
(the relative position causing an error in a record  =  *)
 DAILY ***** OBSERVED *** at ARGONNE
 ?
 Error occurs at or near line *** of (underscore)getday(underscore)
 Called from or near line *** of (underscore)MAIN(underscore)

Cause: This message appears if the first three lines from the data file(s) are not removed.  Delete the first 
three lines of the data file(s) and run LXPET again. 

Problem: While entering the names of the data files, the following error results:

 The value of the STATUS specifier in an OPEN statement does not match the file status (unit=**).  Error 
occurs at or near line ** of (underscore)MAIN(underscore)

Cause: This error occurs when the file name entered is not valid.  Check to be sure that all required data 
files are in the same directory as LXPET.  This error also is caused by misspelled file names entered at the LXPET 
prompt.  Also make sure the file name was not misspelled when it was saved in GenScn.

Appendix B:  LXPET Fortran code 

User Rights Notice

This software and related material (data and (or) documentation), are made available by the U.S. Geological 
Survey (USGS) to be used in the public interest and in the advancement of science.  You may, without any fee or 
cost, use, copy, modify, or distribute this software, and any derivative works thereof, and its supporting documenta-
tion, subject to the following restrictions and understandings.

If you distribute copies or modifications of the software and related material, make sure the recipients receive a 
copy of this notice and receive or can get a copy of the original distribution.  If the software and (or) related mate-
rial are modified and distributed, it must be made clear that the recipients do not have the original and they must be 
informed of the extent of the modifications.  For example, modified files must include a prominent notice stating the 
modifications made, the author of the modifications, and the date the modifications were made. This restriction is 
necessary to guard against problems introduced in the software by others, reflecting negatively on the USGS.

The software is public property and you, therefore, have the right to the source code, if desired.
You may charge fees for distribution, warranties, and services provided in connection with the software or deriv-

ative works thereof.  The name USGS can be used in any advertising or publicity to endorse or promote any products 
or commercial entity using this software if specific written permission is obtained from the USGS.

The user agrees to appropriately acknowledge the authors and the USGS in publications that result from the use 
of this software or in products that include this software in whole or in part.

Because the software and related material are free (other than nominal materials and handling fees) and pro-
vided “as is,” USGS has made no warranty, express or implied, as to accuracy or completeness and is not obligated 
to provide the user with any support, consulting, training or assistance of any kind with regard to the use, operation, 
and performance of this software nor to provide the user with any updates, revisions, new versions, or “bug fixes.”

The user assumes all risk for any damages whatsoever resulting from loss of use, data, or profits arising in con-
nection with the access, use, quality, or performance of this software.
  
C/*                      UTILITY PROCEDURE LXPET                  */
C/*  PURPOSE:                                                     */
C/*    THE LXPET PROCEDURE COMPUTES DAILY EVAPORATION IN 1000THS   */
C/*    OF INCHES FROM DAILY WIND MOVEMENT (MPD), RADIATION        */
C/*    (LANGLEYS/DAY), MAX-MIN AIR TEMPERATURE (DEGREES FAHREN-   */
C/*    HEIT), AND DEWPOINT (DEGREES FAHRENHEIT) USING THE METHOD  */
C/*    OF KOHLER, NORDENSEN, AND FOX (1955) AS ADAPTED TO         */
C/*    COMPUTERS BY LAMOREUX (1962).                              */
C/*
C/*    MODIFIED TO READ AVERAGE AIR TEMP, IT IS READ INTO MAX AIR */
C/*    ARRAY, THE MIN AIR ARRAY ISN’T USED IN THIS VERSION        */
C/*    CALLS DISTRIB.FOR TO DISTRIB THE DAILY DATA INTO HOURLY    */
C/*    DATA                                                       */

      INTEGER YR,MTH,DAY
      INTEGER I,J
      INTEGER RFILE,MXFILE,MNFILE,DFILE,WFILE,EFILE
      INTEGER RESULT
      INTEGER B3,B6

      REAL A,A11,A12,A13,A2,A3,A31,A4,A5
      REAL B,B1,B2,B4,B5
      REAL C,C1,C2,C3
      REAL E,TA

      REAL RAD,TMAX,DEWP,WIND,EVAP
      REAL RFAC,MXFAC,DFAC,WFAC



      CHARACTER*64 RFNAME,MXFNAME,DFNAME,WFNAME,EFNAME

C**********************************************************************
C*    SETUP CONSTANTS                                                 *
C**********************************************************************
      A = 0.0
      A11 = 212.0
      A12 = 0.1024
      A13 = 0.01066
      A2 = 0.0001
      A3 = 0.0105
      A31 = 0.88
      A4 = 0.37
      A5 = 0.0041
      B = 0.0
      B1 = 0.015
      B2 = 398.36
      B3 = -2
      B4 = 68554000000.0
      B5 = -7482.6
      B6 = -1
      C = 0.0
      C1 = 6413300.0
      C2 = 7482.6
      C3 = 398.36
      E = 2.71828
      TA = 0.0
      
C**********************************************************************
C*     GET INFORMATION FROM USER                                      *
C**********************************************************************

      WRITE (*,’(1X,A)’) ‘ENTER SOLAR RADIATION FILENAME:’
      READ  (*,’(A)’) RFNAME
      RFILE=10
      OPEN  (RFILE,FILE=RFNAME,STATUS=’OLD’)
      WRITE (*,’(1X,A)’) 
     $      ‘ENTER MULTIPLICATION FACTOR FOR SOLAR RADIATION:’
      READ  (*,’(F10.0)’) RFAC
      IF (RFAC .EQ. 0) RFAC=1.0

      WRITE (*,’(1X,A)’) ‘ENTER AVERAGE TEMPERATURE FILENAME : ‘
      READ  (*,’(A)’) MXFNAME
      MXFILE=11
      OPEN  (MXFILE,FILE=MXFNAME,STATUS=’OLD’)
      WRITE (*,’(1X,A)’) 
     $      ‘ENTER MULTIPLICATION FACTOR FOR AVERAGE TEMPERATURE:’
      READ  (*,’(F10.0)’) MXFAC
      IF (MXFAC .EQ. 0) MXFAC=1.0
      

      WRITE (*,’(1X,A)’) ‘ENTER DEWPOINT FILENAME            : ‘
      READ  (*,’(A)’) DFNAME
      DFILE=13
      OPEN  (DFILE,FILE=DFNAME,STATUS=’OLD’)
      WRITE (*,’(1X,A)’) 

     $      ‘ENTER MULTIPLICATION FACTOR FOR DEWPOINT:’
      READ  (*,’(F10.0)’) DFAC
      IF (DFAC .EQ. 0) DFAC=1.0
      
      WRITE (*,’(1X,A)’) ‘ENTER WIND FILENAME                : ‘
      READ  (*,’(A)’) WFNAME
      WFILE=14
      OPEN  (WFILE,FILE=WFNAME,STATUS=’OLD’)
      WRITE (*,’(1X,A)’) 
     $      ‘ENTER MULTIPLICATION FACTOR FOR WIND:’
      READ  (*,’(F10.0)’) WFAC
      IF (WFAC .EQ. 0) WFAC=1.0

      WRITE (*,’(1X,A)’) ‘ENTER EVAPORATION FILENAME         : ‘
      READ  (*,’(A)’) EFNAME
      EFILE=15
      OPEN  (EFILE,FILE=EFNAME,STATUS=’UNKNOWN’)

C**********************************************************************
C*    READ IN A DAY OF DATA FROM DATA FILES                          *
C*    SOLAR RADIATION FILE DETERMINES WHICH MONTH/DAY WE’RE ON       *
C**********************************************************************
100   CONTINUE
C +++ RADIATION DATA +++
      YR = 0
      MTH = 0
      DAY = 0
      CALL GETDAY (RFILE,YR,MTH,DAY,RAD,RFAC,RESULT)
      IF (RESULT.EQ.3) GOTO 900
      IF (RESULT.NE.0) THEN
        WRITE (*,’(1X,A)’) ‘RADIATION FILE : ‘
        GOTO 800
      ENDIF

C +++ AVERAGE TEMPERATURE DATA +++
      CALL GETDAY (MXFILE,YR,MTH,DAY,TA,MXFAC,RESULT)
      IF (RESULT.NE.0) THEN
        WRITE (*,’(1X,A)’) ‘AVERAGE TEMPERATURE FILE : ‘
        GOTO 800
      ENDIF

C +++ DEWPOINT DATA +++
      CALL GETDAY (DFILE,YR,MTH,DAY,DEWP,DFAC,RESULT)
      IF (RESULT.NE.0) THEN
        WRITE (*,’(1X,A)’) ‘DEWPOINT FILE : ‘
        GOTO 800
      ENDIF

C +++ WIND DATA +++
      CALL GETDAY (WFILE,YR,MTH,DAY,WIND,WFAC,RESULT)
      IF (RESULT.NE.0) THEN
        WRITE (*,’(1X,A)’) ‘WIND FILE : ‘
        GOTO 800
      ENDIF

C**********************************************************************

C*    CALCULATE EVAP                                                  *
C**********************************************************************

        C = C1*(E**(-C2/(TA+C3))-E**(-C2/(DEWP+C3)))
        IF (C .LT. 0.00001) C = 0.00001
        B = (B1+((TA+B2)**B3)*B4*E**(B5/(TA+B2)))**B6
        A = E**((TA-A11)*(A12-A13*LOG(RAD)))
     $       -A2 + A3*(C**A31)*(A4+A5*WIND)
        EVAP=B*A*1000.0
        IF (EVAP .LT. 0.0) EVAP=0.0
      CALL DISTRIB(EFILE,YR,MTH,DAY,EVAP)

      GOTO 100

800   CONTINUE
      WRITE (*,2010) RESULT
2010  FORMAT (‘ ERROR : ‘,I2)

900   CONTINUE
      CLOSE (RFILE)
      CLOSE (MXFILE)
      CLOSE (DFILE)
      CLOSE (WFILE)
      CLOSE (EFILE)
      END

C**********************************************************************
C*    SUBROUTINE GETDAY / GET A DAY OF DATA FROM FILE                 *
C**********************************************************************
      SUBROUTINE GETDAY(FILE,TY,TM,TD,DATA,MFAC,RESULT)

C +++ ARGUMENT VARIABLES +++
      INTEGER FILE,TY,TM,TD,RESULT
      REAL DATA,MFAC
   
C +++ LOCAL VARIABLES +++

      INTEGER YR,MTH,DAY,I

1000  FORMAT (I4,1X,I2,1X,I2,20X,F10.2) 
      RESULT = 0
100   CONTINUE
      READ (FILE,1000,END=800) YR,MTH,DAY,DATA

C +++ CHECK TO SEE IF WE GOT THE MONTH/DAY WE WANT +++
      IF ((0.EQ.TY) .AND. (0.EQ.TM) .AND. (0.EQ.TD)) THEN
        TY = YR
        TM = MTH
        TD = DAY
        GOTO 900
      ENDIF

      IF (YR.EQ.TY) THEN
        IF (MTH.EQ.TM) THEN
           IF (DAY.EQ.TD) GOTO 900

        IF (MTH.LT.TM) GOTO 100
        ENDIF
        RESULT=2
        GOTO 900
      ENDIF

      IF (YR .LT. TY) GOTO 100
      RESULT = 2
      GOTO 900

800   CONTINUE
C +++ UNEXPECTED END OF FILE +++
      RESULT = 3

900   CONTINUE
      IF (MFAC .NE. 1.0) THEN
          DATA=MFAC*DATA
      ENDIF
      RETURN
      END

C *****************************************************************
C * SUBROUTINE DISTRIB     *
C *  CONVERT DAILY EVAP DATA TO HOURLY EVAP DATA BASED *
C *   ON DISTRIBUTION FROM LANDS PROGAM *
C *****************************************************************
      SUBROUTINE DISTRIB(EFILE,YR,MTH,DAY,DAYDAT)
      REAL HRDAT(24),DAYDAT,EVAPDIST(24)
      INTEGER YR,MTH,DAY,I,J,EFILE

      DATA EVAPDIST / 0.000,0.000,0.000,0.000,0.000,0.000,0.019,0.041,
     $                0.067,0.088,0.102,0.110,0.110,0.110,0.105,0.095,
     $                0.081,0.055,0.017,0.000,0.000,0.000,0.000,0.000/

      DO 200,J=1,24
        HRDAT(J)=DAYDAT*EVAPDIST(J)
200   CONTINUE

      WRITE(EFILE,2000) YR,MTH,DAY,1,(HRDAT(J),J=1,12)
      WRITE(EFILE,2000) YR,MTH,DAY,2,(HRDAT(J),J=13,24)
2000  FORMAT(8X,I4,1X,2(I2,1X),I1,12F5.2)
      RETURN
      END

